

# HP Prime VCR Plus+ Decode/Encode Program

```
// HP Prime VCRPlus+ Decode/Encode
// Program will handle 1-8 digit codes
// List of exported functions
PLUS(); // Decode/Encode VCRPlus+ code
DECO(); // Decode newspaper code
FUN1(); // Unmask newspaper code
MTOPI(); // Process upper 3-5 digits
SHUF(); // Shuffle bits > time, channel
TWID(); // Adjust start time and length
ENCO(); // Encode newspaper code
ITWD(); // Start time, length -> table
ITLV(); // Interleave time/channel bits
IMTP(); // Calculate upper 3-5 digits
IFUN(); // Create final newspaper code
DOTW(); // Day of the week subroutine
LIST(); // Table lists, channel names

EXPORT PLUS() // Call Decode or Encode
BEGIN // Make local variables
LOCAL n, st, sh, s1, s2, sm, t1, t2;
LOCAL ft, fh, f1, f2, fm;
LOCAL code, day, month, year;
LOCAL i, x, y, z, x1, x2, y1, y2;

HSeparator:=3; // No digit separator
x1:=Xmin; // Save existing screen
x2:=Xmax; // configuration; might
y1:=Ymin; // be different for the
y2:=Ymax; // various applications
x:=RGB(40, 100, 175); // Dark blue top
y:=RGB(178, 255, 255); // Lite blue box
z:=RGB(255, 255, 255); // White letters

IF SIZE(L1) <> 192 OR SIZE(L2) <> 192
THEN LIST(); // Create L1-L5
END; // if not found

CHOOSE(n, "Choose", "Decode", "Encode");
n:=when(n == 1, DECO(), ENCO());
code:=n(1); // Parse array n
month:=n(2); // returned from
day:=n(3); // Decode|Encode
year:=n(4); // 1440 mins/day
st:=n(5) MOD 1440; // Starting time
sh:=(IP(st / 60) + 23) MOD 12 + 1;
s1:=IP(st MOD 60 / 10);
s2:=st MOD 10;
sm:=when(st < 720, "AM", "PM");
```

## HP Prime VCR Plus+ Decode/Encode Program

```
ft:=(n(5) + n(6)) MOD 1440; // End time
fh:=(IP(ft / 60) + 23) MOD 12 + 1;
f1:=IP(ft MOD 60 / 10);
f2:=ft MOD 10;
fm:=when(ft < 720, "AM", "PM");
t1:=IP(n(6) / 60); // Length hours
t2:=n(6) MOD 60; // Length mins

Xmin:=1; // Set screen size
Xmax:=319; // 320 pixels wide
Ymin:=1; // by
Ymax:=219; // 220 pixels tall
RECT(); // Clear screen
RECT(0, 220, 320, 201, 0, x); // Top
FOR i FROM 0 TO 5 DO // Draw 12 boxes
  RECT( 5, 190 - i * 20, 65, 175 - i * 20, 0, y);
  RECT( 75, 190 - i * 20, 195, 175 - i * 20, 0, z);
END;

// Populate boxes with text and data
TEXTOUT("VCRPLUS+", 125, 218, 3, z);
TEXTOUT("Code", 10, 190);
TEXTOUT("Date", 10, 170);
TEXTOUT("Start", 10, 150);
TEXTOUT("Finish", 10, 130);
TEXTOUT("Length", 10, 110);
TEXTOUT("Channel", 10, 90);
TEXTOUT(n(1), 80, 190);
TEXTOUT(n(2)+"/"+n(3)+"/"+n(4)+" "+DOTW(n(2),n(3),n(4)), 80, 170);
TEXTOUT(sh+": "+s1+s2+" "+sm, 80, 150);
TEXTOUT(fh+": "+f1+f2+" "+fm, 80, 130);
TEXTOUT(t1+" hr "+t2+" min", 80, 110);
TEXTOUT(n(7)+" - "+L5(n(7)), 80, 90);
FREEZE; // Stop screen and display

Xmin:=x1; // Restore original screen
Xmax:=x2; // configuration of whatever
Ymin:=y1; // application was selected
Ymax:=y2; // prior to program running
END;
```

## HP Prime VCR Plus+ Decode/Encode Program

```
EXPORT DECO() // Decode newspaper code
BEGIN
LOCAL year, month, day, code, n, x;
INPUT({{year, [0], {15, 25, 0}},
      {month, [0], {15, 25, 1}},
      {code, [0], {15, 25, 2}}},
      "VCRPLUS+",
      {"Year", "Month", "Code"},
      {"Enter Year 1989-2089",
       "Enter Month 1-12",
       "Enter Code 1-99999999"});
year:=MIN(MAX(IP(year), 1989), 2089);
month:=MIN(MAX(IP(month), 1), 12);
code:=MIN(MAX(IP(code), 1), 99999999);
x:=FUN1(code);
day:=IP((x MOD 1000 - 1) / 32) + 1;
n:=MTOP(year, month, day, IP(x / 1000), (x MOD 1000 - 1) MOD 32);
n:=SHUF(n(1), n(2));
x:=n(2);
n:=TWID(n(1));
RETURN {code, month, day, year, n(1), n(2), x + 1};
END;
```

```
EXPORT FUN1(code) // Unmask printed code
BEGIN
LOCAL i, j, n, nd;
n:=REVERSE(ASC(STRING(code))-48);
nd:=SIZE(n);
REPEAT
  FOR i FROM 1 TO 3 DO
    FOR j FROM 2 TO nd DO
      n(j):=(n(j - 1) + n(j)) MOD 10;
    END;
  END;
UNTIL n(nd) <> 0 OR code == 0;
RETURN EXPR(CHAR(REVERSE(n)+48));
END;
```

## HP Prime VCR Plus+ Decode/Encode Program

```
// Process top 3-5 digits of code
EXPORT MTOP(year, month, day, top5, rem)
BEGIN
LOCAL i, j, n, nd, t1, t2, t3, ym;
n:=REVERSE(ASC(STRING(top5))-48);
nd:=SIZE(n);
rem:=(rem + sum(n)) MOD 32;
IF nd <= 3 THEN
  if top5 > 0 THEN
    ym:=year MOD 100 MOD 16;
    REPEAT
      FOR i FROM 0 TO ym DO
        n(nd):=(n(nd) + day) MOD 10;
        FOR j FROM nd - 1 DOWNTO 1 DO
          n(j):=(n(j) + n(j + 1)) MOD 10;
        END;
        rem:=rem + n(1);
      END;
    UNTIL n(nd) <> 0;
  END;
  rem:=(rem + (day * (month + 1))) MOD 32;
ELSE
  ym:=year MOD 100 * 12 + month;
  REPEAT
    FOR i FROM 1 TO 31 DO
      t1:=(ym + 31 - i) MOD 31 + 1;
      t1:=when(nd == 4, L3(t1), L4(t1));
      t2:=IP(t1 / 16) + 1;
      t3:=(t1 MOD 16) + 1;
      t1:=10 * n(t2) + n(t3);
      REPEAT
        t1:=(L1(t1 + 1) - ym + 1920) MOD 192;
      UNTIL t1 < 100;
      n(t2):=IP(t1 / 10);
      n(t3):=(t1 MOD 10);
    END;
  UNTIL n(nd) <> 0;
END;
RETURN {EXPR(CHAR(REVERSE(n)+48)), rem};
END;
```

## HP Prime VCR Plus+ Decode/Encode Program

```
// Shuffle bits into time, channel info
EXPORT SHUF(top5, rem)
BEGIN
LOCAL cc, tt;
tt:=BITAND(rem,      1) +
    BITAND(rem,      4) / 2 +
    BITAND(rem,     16) / 4 +
    BITAND(top5,     1) * 8 +
    BITAND(top5,    56) * 2 +
    BITAND(top5,   128) +
    BITAND(top5,  3584) / 2 +
    BITAND(top5, 57344) / 4;
cc:=BITAND(rem,      2) / 2 +
    BITAND(rem,      8) / 4 +
    BITAND(top5,     6) * 2 +
    BITAND(top5,    64) / 4 +
    BITAND(top5,   256) / 8 +
    BITAND(top5,  4096) / 64 +
    BITAND(top5, 65536) / 512;
IF BITAND(top5, 65536) THEN
    tt:=BITAND(tt,  51199) +
        BITAND(top5, 28672) / 2;
    cc:=BITAND(cc,  191) +
        BITAND(top5, 32768) / 512;
END;
RETURN {tt, cc};
END;
```

## HP Prime VCR Plus+ Decode/Encode Program

```
// Adjust start time and length
EXPORT TWID(tt)
BEGIN
LOCAL d, t, t1, t2, x1, x2;
CASE
  IF tt >= 768 AND tt <= 3647 THEN
    t1:=(tt - 768) MOD 10 + 1;
    t2:=t1 * 5;
    IF t1 >= 6 THEN
      x2:=t2 - 25;
    ELSE
      x1:=t2;
    END;
    tt:=IP((tt - 768) / 10);
  END;
  IF tt >= 3648 AND tt <= 6527 THEN
    t1:=(tt - 3648) MOD 6 + 1;
    IF t1 == 1 THEN
      x1:=15;
    ELSE
      x2:=t1 * 5 - 5;
    END;
    tt:=IP((tt - 3648) / 6) + 288;
  END;
  IF tt >= 6528 AND tt <= 13727 THEN
    x1:=5 + (tt - 6528) MOD 25 MOD 5 * 5;
    x2:=5 + IP((tt - 6528) MOD 25 / 5) * 5;
    tt:=IP((tt - 6528) / 25);
  END;
  IF tt >= 13728 THEN
    x1:=15;
    x2:=5 + (tt - 13728) MOD 5 * 5;
    tt:=IP((tt - 13728) / 5) + 288;
  END;
END;
IF tt < 192 THEN
  t:=L1(tt + 1) MOD 48;
  d:=IP(L1(tt + 1) / 48);
ELSE
  t:=47 - (tt - 192) MOD 48;
  d:=IP(tt / 48);
END;
RETURN {t * 30 + x2, d * 30 + 30 - x1};
END;
```

## HP Prime VCR Plus+ Decode/Encode Program

```
EXPORT ENCO() // Encode newspaper code
BEGIN
LOCAL year, month, day, code, start, durn, chan, n;
INPUT({{year, [0], {20, 25, 0}},
      {month, [0], {20, 25, 1}},
      {day, [0], {20, 25, 2}},
      {start, [0], {20, 25, 3}},
      {durn, [0], {20, 25, 4}},
      {chan, [0], {20, 25, 5}}},
      "VCRPLUS+",
      {"Year", "Month", "Day",
      "Start", "Length", "Channel"},
      {"Enter Year 1989-2089",
      "Enter Month 1-12",
      "Enter Day 1-31",
      "Enter Start HHMM 0-2355",
      "Enter Length HMM 5-800",
      "Enter Channel 1-192"});
year:=MIN(MAX(IP(year), 1989), 2089);
month:=MIN(MAX(IP(month), 1), 12);
day:=MIN(MAX(IP(day), 1), 31);
start:=MIN(MAX(IP(start), 0), 2355);
start:=IP(start / 100) * 60 + start MOD 100;
start:=MIN(MAX(IP(start / 5) * 5, 0), 1435);
durn:=MIN(MAX(IP(durn), 0), 800);
durn:=IP(durn / 100) * 60 + durn MOD 100;
durn:=MIN(MAX(IP((durn + 4) / 5) * 5, 5), 480);
IF durn > 180 AND durn MOD 15 THEN
  durn:=(IP(durn / 15) + 1) * 15;
END;
chan:=MIN(MAX(IP(chan), 1), 192);
n:=ITWD(start, durn);
n:=ITLV(n, chan - 1);
n:=IMTP(year, month, day, n(1), n(2));
code:=IFUN(n(1) * 1000 + n(2) + 1 + (32 * (day - 1)));
RETURN {code, month, day, year, start, durn, chan};
END;
```

## HP Prime VCR Plus+ Decode/Encode Program

```
// Start time, length -> table number
EXPORT ITWD(startm, durnm)
BEGIN
LOCAL basett, d, t, tt, x1, x2;
startm:=MIN(MAX(IP(startm / 5) * 5, 0), 1435);
durnm:=MIN(MAX(IP((durnm + 4) / 5) * 5, 5), 480);
IF durnm > 180 AND durnm MOD 15 THEN
    durnm:=(IP(durnm / 15) + 1) * 15;
END;
t:=IP(startm / 30);
d:=IP((durnm + 29) / 30);
x1:=d * 30 - durnm;
x2:=startm - t * 30;
IF d <= 4 THEN
    basett:=L2(t + (d - 1) * 48 + 1);
ELSE
    basett:=48 * (d - 1) + 47 - t;
END;
CASE
    IF x1 == 0 AND x2 == 0 THEN
        tt:= basett;
    END;
    IF d <= 6 AND (x1 == 0 OR x2 == 0) THEN
        tt:= 768 + 10 * basett;
        IF x1 THEN
            tt:=tt + x1 / 5 - 1;
        END;
        IF x2 THEN
            tt:=tt + x2 / 5 + 4;
        END;
    END;
    IF d > 6 AND ((x1 == 15 AND x2 == 0) OR (x1 == 0 AND x2 <> 0)) THEN
        tt:=3648 + 6 * (basett - 288);
        tt:=tt + x2 / 5;
    END;
    IF d <= 6 AND x1 AND x2 THEN
        tt:=6528 + 25 * basett;
        tt:=tt + x1 / 5 - 1;
        tt:=tt + (x2 / 5 - 1) * 5;
    END;
    IF d > 6 AND x1 AND x2 THEN
        tt:=13728 + 5 * (basett - 288);
        tt:=tt + x2 / 5 - 1;
    END;
    tt:=-1;
END;
RETURN tt;
END;
```

## HP Prime VCR Plus+ Decode/Encode Program

```
// Interleave time and channel bit info
EXPORT ITLV(tval, cval)
BEGIN
LOCAL rem, top5;
rem:= BITAND(tval, 1) +
      BITAND(cval, 1) * 2 +
      BITAND(tval, 2) * 2 +
      BITAND(cval, 2) * 4 +
      BITAND(tval, 4) * 4;
top5:=BITAND(tval, 8) / 8 +
      BITAND(cval, 12) / 2 +
      BITAND(tval, 112) / 2 +
      BITAND(cval, 16) * 4 +
      BITAND(tval, 128) +
      BITAND(cval, 32) * 8 +
      BITAND(tval, 1792) * 2 +
      BITAND(cval, 64) * 64 +
      BITAND(tval, 14336) * 4 +
      BITAND(cval, 128) * 512;
IF BITAND(top5, 65536) THEN
  top5:=BITAND(top5, 69631) +
        BITAND(tval, 14336) * 2 +
        BITAND(cval, 64) * 512;
END;
RETURN {top5, rem};
END;
```

## HP Prime VCR Plus+ Decode/Encode Program

```
// Calculate upper 3-5 digits of code
EXPORT IMTP(year, month, day, top5, rem)
BEGIN
LOCAL i, j, n, nd, t1, t2, t3, ym;
n:=REVERSE(ASC(STRING(top5))-48);
nd:=SIZE(n);
IF nd <= 3 THEN
  rem:=(rem + 640 - (day * (month + 1))) MOD 32;
  IF top5 > 0 THEN
    ym:=year MOD 100 MOD 16;
    REPEAT
      FOR i FROM 0 TO ym DO
        rem:=rem - n(1);
        FOR j FROM 1 TO nd - 1 DO
          n(j):=(n(j) + 100 - n(j + 1)) MOD 10;
        END;
        n(nd):=(n(nd) + 100 - day) MOD 10;
      END;
    UNTIL n(nd) <> 0;
  END;
ELSE
  ym:=year MOD 100 * 12 + month;
  REPEAT
    FOR i FROM 31 DOWNTO 1 DO
      t1:=(ym + 31 - i) MOD 31 + 1;
      t1:=when(nd == 4, L3(t1), L4(t1));
      t2:=IP(t1 / 16) + 1;
      t3:=(t1 MOD 16) + 1;
      t1:=10 * n(t2) + n(t3);
      REPEAT
        t1:=L2(((t1 + ym) MOD 192) + 1);
      UNTIL t1 < 100;
      n(t2):=IP(t1 / 10);
      n(t3):=(t1 MOD 10);
    END;
  UNTIL n(nd) <> 0;
END;
rem:=(rem + 640 - sum(n)) MOD 32;
RETURN {EXPR(CHAR(REVERSE(n)+48)), rem};
END;
```

## HP Prime VCR Plus+ Decode/Encode Program

```
// Create final newspaper code
EXPORT IFUN(code)
BEGIN
LOCAL i, j, n, nd;
n:=REVERSE(ASC(STRING(code))-48);
nd:=SIZE(n);
REPEAT
  FOR i FROM 1 TO 3 DO
    FOR j FROM nd DOWNTO 2 DO
      n(j):=(n(j) + 10 - n(j - 1)) MOD 10;
    END;
  END;
UNTIL n(nd) <> 0 OR code == 0;
RETURN EXPR(CHAR(REVERSE(n)+48));
END;
```

```
// Month, day, year -> day of the week
EXPORT DOTW(month, day, year)
BEGIN
month:=(month - 3) MOD 12 + 4;
year:=year - IP(month / 14);
RETURN MID("SATSUNMONTUEWEDTHUFRI",
  (IP(2.6 * month) + IP(1.25 * year) -
  IP(year / 100) + IP(year / 400) +
  day) MOD 7 * 3 + 1, 3);
END;
```

# HP Prime VCR Plus+ Decode/Encode Program

```
// Create time tables and channel names
EXPORT LIST()
BEGIN

// L1 used for decoding - DO NOT CHANGE

L1:={
37,32,39,33,31,35,36,29,38,82,28,41,34,176,40,30,
184,186,88,180,86,92,90,172,78,188,23,22,46,80,138,42,
25,27,19,74,43,72,164,84,44,24,16,17,178,18,45,21,
182,15,94,68,14,26,158,70,76,20,160,47,170,168,162,13,
132,12,11,0,191,140,122,66,129,128,125,136,133,60,120,1,
147,48,130,144,64,62,187,10,127,167,166,113,141,114,139,177,
95,146,124,3,151,126,174,190,134,112,9,6,171,116,110,2,
143,7,4,189,8,156,152,5,109,49,93,50,145,142,81,65,
96,183,163,137,58,83,148,135,115,179,157,85,77,119,97,161,
117,173,98,159,185,102,150,123,121,101,91,71,181,61,59,52,
175,63,108,131,56,111,105,57,99,169,51,149,87,54,165,100,
55,154,67,53,89,104,79,153,75,73,103,69,106,155,107,118
};

// L2 used for encoding - DO NOT CHANGE

L2:={
67,79,111,99,114,119,107,113,116,106,87,66,65,63,52,49,
42,43,45,34,57,47,27,26,41,32,53,33,10,7,15,4,
1,3,12,5,6,0,8,2,14,11,31,36,40,46,28,59,
81,121,123,170,159,179,173,176,164,167,132,158,77,157,85,161,
84,127,71,178,51,187,55,155,37,185,35,184,56,140,24,182,
29,126,9,133,39,139,20,172,18,180,22,154,21,122,50,96,
128,142,146,168,175,153,149,186,181,166,188,190,162,120,110,165,
105,91,93,136,109,144,191,141,78,152,70,151,98,74,101,88,
73,72,82,163,64,76,104,135,75,131,30,94,69,92,125,112,
83,124,97,80,134,171,150,100,118,183,177,189,117,138,54,147,
58,143,62,130,38,174,90,89,61,169,60,108,23,145,102,160,
13,95,44,137,19,156,48,129,16,148,17,86,25,115,103,68
};

// L3 used for 7 digits - DO NOT CHANGE

L3:={
3,49,18,18,48,3,3,33,
35,18,33,49,3,48,1,33,
49,16,19,48,18,3,18,48,
49,19,33,19,48,49,48
};
```

# HP Prime VCR Plus+ Decode/Encode Program

```
// L4 used for 8 digits - DO NOT CHANGE
```

```
L4:={  
20,2,35,3,65,20,20,50,  
52,3,50,64,16,1,18,50,  
64,33,36,65,35,20,35,65,  
2,4,48,36,65,64,1  
};
```

```
// L5 is channel names - OK TO CHANGE
```

```
L5:={  
"WXXA1", "WXXB2", "WXXC3", "WXXD4", "WXXE5", "WXXF6",  
"WXXG7", "WXXH8", "WXXI9", "WXXJ10", "WXXK11", "WXXL12",  
"WXXM13", "WXXN14", "WXXO15", "WXXP16", "WXXQ17", "WXXR18",  
"WXXS19", "WXXT20", "WXXU21", "WXXV22", "WXXW23", "WXXX24",  
"WXXY25", "WXXZ26", "NGC", "NASA", "AXS", "C&I", "FOX", "OWN",  
"HBO", "ESPN", "AMC", "ION", "DSC", "NIK", "A&E", "FOOD", "SHO",  
"CNN", "TBS", "USA", "MAX", "LIF", "FAM", "MTV", "TNN", "MAG",  
"TLC", "TNT", "DIS", "BRV", "OVA", "VER", "BET", "TMC", "FSN",  
"AM1", "BBTV", "VH1", "E!", "DFAM", "DJR", "DXD", "HMK", "QUBO",  
"H&G", "APL", "UP", "CIN", "HIS", "EPX", "COM", "TCM", "FLIX",  
"IFC", "CNBC", "STARZ", "MGM", "LMN", "SONY", "SUN", "FX", "CRT",  
"MZK", "ES2", "SCI", "MSNBC", "FNC", "WBM", "CMT", "FUSE", "HAR",  
"PAL", "SPOT", "TCT", "QVC", "CSPAN", "HLN", "REN", "HAL", "CSN",  
"ALJ", "NOG", "AWC", "EURO", "SNN", "TWC", "TXCN", "NY1", "ONTV",  
"HRTV", "PBS", "MLB", "NHL", "SKI", "TEAM", "NBA", "NFL", "TVG",  
"CSS", "COX", "MSG", "ROOT", "YES", "WGN", "WXYZ", "KTLA", "TUX",  
"WLS", "WWJ", "WBZ", "KNBC", "KUSA", "CTN", "EWTN", "BYU", "NRB",  
"TBN", "TRI", "GEM", "HSN", "SHQ", "GSN", "AAN", "HITN", "MEX",  
"LATV", "TRES", "UNI", "MEGA", "ARTN", "VIA", "RSC", "MKTV",  
"SPT", "MYX", "JUS", "TKC", "SKY", "ITV", "DFH", "TVK", "HOT",  
"ARTS", "ATN", "G4", "HA!", "MOJO", "CTS", "NJTN", "FRC", "TRIO",  
"TSN", "SOAP", "PLAY", "CHC", "FLN", "KAL", "STYLE", "NOYZ",  
"TUBE", "ZMT", "UNO", "BOX", "APC", "NWI", "SNC", "PPV", "BFC"  
};
```

```
END;
```

```
// End of program code
```